

Soluções de EDOs em Séries de Potências - solução em torno de um ponto singular

Fernando Deeke Sasse
Departamento de Matemática
CCT - UDESC

1. Consideremos o problema de resolver a equação

$$x \frac{d^2}{dx^2} y(x) + 3 \frac{d}{dx} y(x) - y(x) = 0$$

em torno do ponto singular regular $x = 0$.

Definamos separadamente cada termo da equação:

```
> restart;  
> P := x: Q:=3: R:=-1:  
> S2 := sum(simplify(P*c[n]*(n+r)*(n+r-1)*x^(n+r-2)), n = 0 .. k);
```

$$S2 := \sum_{n=0}^k x^{n+r-1} c_n (n+r) (n+r-1) \quad (1)$$

```
> S1 := sum(simplify(Q*c[n]*(n+r)*x^(n+r-1)), n = 0 .. k);
```

$$S1 := \sum_{n=0}^k 3 x^{n+r-1} c_n (n+r) \quad (2)$$

```
> S0 := sum(simplify(R*c[n]*x^(n+r)), n = 0 .. k);
```

$$S0 := \sum_{n=0}^k (-c_n x^{n+r}) \quad (3)$$

A equação pode agora ser escrita como

```
> S := S2+S1+S0;
```

(4)

$$S := \sum_{n=0}^k x^{n+r-1} c_n (n+r) (n+r-1) + \sum_{n=0}^k 3 x^{n+r-1} c_n (n+r) + \sum_{n=0}^k (-c_n x^{n+r}) \quad (4)$$

Consideremos expansões até quarta ordem:

```
> k := 4;
```

$$k := 4 \quad (5)$$

Eliminando o fator comum x^r temos

```
> eq1 := simplify(S/x^r);
```

$$eq1 := \frac{1}{x} (-c_4 x^5 - c_3 x^4 + c_1 r^2 x - c_1 x^2 + 3 c_1 x - c_2 x^3 - c_0 x + c_0 r^2 + 4 c_1 r x + 6 x^2 c_2 r + x^2 c_2 r^2 + 8 x^3 c_3 r + x^3 c_3 r^2 + 10 x^4 c_4 r + x^4 c_4 r^2 + 2 c_0 r + 8 c_2 x^2 + 15 c_3 x^3 + 24 c_4 x^4) \quad (6)$$

Por simplicidade, vamos multiplicar toda a equação por x :

```
> eq2 := x*%;
```

```
> eq3 := collect(%, x);
```

$$eq3 := -c_4 x^5 + (-c_3 + c_4 r^2 + 10 c_4 r + 24 c_4) x^4 + (-c_2 + 15 c_3 + c_3 r^2 + 8 c_3 r) x^3 + (6 c_2 r + 8 c_2 + c_2 r^2 - c_1) x^2 + (3 c_1 + 4 c_1 r - c_0 + c_1 r^2) x + 2 c_0 r + c_0 r^2 \quad (7)$$

A equação indicial é dada por

```
> e0 := coeff(eq3, x, 0);
```

$$e0 := 2 c_0 r + c_0 r^2 \quad (8)$$

```
> s := solve(e0, r);
```

$$s := 0, -2 \quad (9)$$

```
> r1 := s[1]; r2 := s[2];
```

$$\begin{aligned} r1 &:= 0 \\ r2 &:= -2 \end{aligned} \quad (10)$$

Como $r1 - r2 = 1$ é inteiro, as relações de recorrências associadas a cada raiz não geram soluções linearmente independentes.

Determinemos as relações de recorrência associadas a $r = 0$:

```
> r := r1;
```

```
> eqs := array(1 .. k);
```

```
for n from 1 to k do
```

```
    eqs[n] := coeff(eq3, x, n) = 0
```

```
end do;
```

```
coeffEqs := convert(eqs, list);
```

```
eqs := array(1..4, [ ])
```

```
eqs1 := 3 c1 - c0 = 0
```

```
eqs2 := 8 c2 - c1 = 0
```

```
eqs3 := -c2 + 15 c3 = 0
```

```
eqs4 := -c3 + 24 c4 = 0
```

```
coeffEqs := [3 c1 - c0 = 0, 8 c2 - c1 = 0, -c2 + 15 c3 = 0, -c3 + 24 c4 = 0]
```

(11)

Temos aqui 4 equações para 6 incógnitas c_1 a c_4 .

```
> s1 := [seq(c[n], n = 1 .. k)];
```

```
s1 := [c1, c2, c3, c4]
```

(12)

Resolvendo estas equações para a_0 obtemos

```
> coef := solve(coeffEqs, s1);
```

```
coef := [[c1 =  $\frac{1}{3}$  c0, c2 =  $\frac{1}{24}$  c0, c3 =  $\frac{1}{360}$  c0, c4 =  $\frac{1}{8640}$  c0]]
```

(13)

```
> assign(coef);
```

Uma série que é uma solução particular é, portanto, dada por

```
> y1 := sum(simplify(c[j]*x^(j+r)), j = 0 .. k);
```

```
y1 := c0 +  $\frac{1}{3}$  c0x +  $\frac{1}{24}$  c0x2 +  $\frac{1}{360}$  c0x3 +  $\frac{1}{8640}$  c0x4
```

(14)

Para encontrar a segunda solução podemos usar a fórmula:

$$y_2 = y_1 \int \frac{1}{y_1^2} \exp\left(-\int P(x) dx\right) dx,$$

sendo $P(x)$ o coeficiente da equação diferencial linear homogênea de segunda ordem na forma

$$\frac{d^2}{dx^2} y(x) + P(x) \frac{d}{dx} y(x) + Q(x) y(x) = 0$$

No presente caso,

$$P(x) = \frac{3}{x}$$

de modo que

```
> P:=3/x;
```

$$P := \frac{3}{x}$$

(15)

```
> ex:=exp(-int(P,x));
```

$$ex := \frac{1}{x^3}$$

(16)

Por outro lado,

```
> f:=1/y1^2;
```

$$f := \frac{1}{\left(c_0 + \frac{1}{3} c_0 x + \frac{1}{24} c_0 x^2 + \frac{1}{360} c_0 x^3 + \frac{1}{8640} c_0 x^4\right)^2}$$

(17)

```
> fs:=series(f,x,5);
```

$$fs := \frac{1}{c_0^2} - \frac{2}{3 c_0^2} x + \frac{1}{4 c_0^2} x^2 - \frac{19}{270 c_0^2} x^3 + \frac{433}{25920 c_0^2} x^4 + O(x^5)$$

(18)

```
> fp:=convert(fs,polynom);
```

$$fp := \frac{1}{c_0^2} - \frac{2}{3} \frac{x}{c_0^2} + \frac{1}{4} \frac{x^2}{c_0^2} - \frac{19}{270} \frac{x^3}{c_0^2} + \frac{433}{25920} \frac{x^4}{c_0^2}$$

(19)

A solução y_2 é então dada por

```
> y2:=y1*int(fp*ex,x);
```

$$y_2 := \left(c_0 + \frac{1}{3} c_0 x + \frac{1}{24} c_0 x^2 + \frac{1}{360} c_0 x^3 + \frac{1}{8640} c_0 x^4\right) \left(\frac{433}{51840} \frac{x^2}{c_0^2} - \frac{19}{270} \frac{x}{c_0^2} - \frac{1}{2 c_0^2 x^2} + \frac{2}{3 c_0^2 x} + \frac{1}{4} \frac{\ln(x)}{c_0^2}\right)$$

(20)

Por consistência, devemos manter termos até segunda ordem em x :

```
> y22:=series(y2,x,3);
```

(21)

$$y_{22} := -\frac{1}{2c_0} \frac{1}{x^2} + \frac{1}{2c_0} \frac{1}{x} + \frac{1}{4} \frac{\ln(x)}{c_0} + \frac{29}{144c_0} + O(x) \quad (21)$$

> `y22p:=convert(y22,polynom);`

$$y_{22p} := -\frac{1}{2c_0x^2} + \frac{1}{2c_0x} + \frac{1}{4} \frac{\ln(x)}{c_0} + \frac{29}{144c_0} \quad (22)$$

A solução geral é então dada por

> `Y :=C[1]*y1+C[2]*y22p;`

$$Y := C_1 \left(c_0 + \frac{1}{3} c_0 x + \frac{1}{24} c_0 x^2 + \frac{1}{360} c_0 x^3 + \frac{1}{8640} c_0 x^4 \right) + C_2 \left(-\frac{1}{2c_0x^2} + \frac{1}{2c_0x} + \frac{1}{4} \frac{\ln(x)}{c_0} + \frac{29}{144c_0} \right) \quad (23)$$

O Maple resolve esta equação em termos de funções especiais. Suponhamos condições iniciais $y(0.1) = 1, y'(0.1) = -1$.

> `restart;`

> `ics := y3(0.1) = 1, (D(y3))(0.1) = -1;`

> `eq4 := x*(diff(y3(x), x, x))+3*(diff(y3(x), x))-y3(x)=0;`

$$eq4 := x \left(\frac{d^2}{dx^2} y_3(x) \right) + 3 \left(\frac{d}{dx} y_3(x) \right) - y_3(x) = 0 \quad (24)$$

> `dsolve({eq4, ics})`

$$y_3(x) = \frac{1}{100} \frac{\left(19 \operatorname{BesselK}\left(2, \frac{1}{5} \sqrt{10}\right) + \sqrt{10} \operatorname{BesselK}\left(1, \frac{1}{5} \sqrt{10}\right) \right) \sqrt{10} \operatorname{BesselI}\left(2, 2\sqrt{x}\right)}{\left(\operatorname{BesselK}\left(2, \frac{1}{5} \sqrt{10}\right) \operatorname{BesselI}\left(1, \frac{1}{5} \sqrt{10}\right) + \operatorname{BesselK}\left(1, \frac{1}{5} \sqrt{10}\right) \operatorname{BesselI}\left(2, \frac{1}{5} \sqrt{10}\right) \right) x} - \frac{1}{100} \frac{\sqrt{10} \left(19 \operatorname{BesselI}\left(2, \frac{1}{5} \sqrt{10}\right) - \sqrt{10} \operatorname{BesselI}\left(1, \frac{1}{5} \sqrt{10}\right) \right) \operatorname{BesselK}\left(2, 2\sqrt{x}\right)}{\left(\operatorname{BesselK}\left(2, \frac{1}{5} \sqrt{10}\right) \operatorname{BesselI}\left(1, \frac{1}{5} \sqrt{10}\right) + \operatorname{BesselK}\left(1, \frac{1}{5} \sqrt{10}\right) \operatorname{BesselI}\left(2, \frac{1}{5} \sqrt{10}\right) \right) x} \quad (25)$$

> `ymaple:=op(2,%);`

(26)


$$y_{maple} := \frac{1}{100} \frac{\left(19 \operatorname{BesselK}\left(2, \frac{1}{5} \sqrt{10}\right) + \sqrt{10} \operatorname{BesselK}\left(1, \frac{1}{5} \sqrt{10}\right) \right) \sqrt{10} \operatorname{BesselI}\left(2, 2\sqrt{x}\right)}{\left(\operatorname{BesselK}\left(2, \frac{1}{5} \sqrt{10}\right) \operatorname{BesselI}\left(1, \frac{1}{5} \sqrt{10}\right) + \operatorname{BesselK}\left(1, \frac{1}{5} \sqrt{10}\right) \operatorname{BesselI}\left(2, \frac{1}{5} \sqrt{10}\right) \right) x}$$

$$- \frac{1}{100} \frac{\sqrt{10} \left(19 \operatorname{BesselI}\left(2, \frac{1}{5} \sqrt{10}\right) - \sqrt{10} \operatorname{BesselI}\left(1, \frac{1}{5} \sqrt{10}\right) \right) \operatorname{BesselK}\left(2, 2\sqrt{x}\right)}{\left(\operatorname{BesselK}\left(2, \frac{1}{5} \sqrt{10}\right) \operatorname{BesselI}\left(1, \frac{1}{5} \sqrt{10}\right) + \operatorname{BesselK}\left(1, \frac{1}{5} \sqrt{10}\right) \operatorname{BesselI}\left(2, \frac{1}{5} \sqrt{10}\right) \right) x}$$

(26)

Façamos o gráfico desta solução no intervalo [0.1, 3].

```
> with(plots):
> xf:=3:
> Pmaple := plot(ymaple, x=0.05..xf, legend=sol.Bessel):
> display(Pmaple);
```



Resolvamos este problema através de cálculo numérico.

```
> ynum:=dsolve({eq4,ics}, numeric);
                                ynum:=proc(x_rkf45) ... end proc
> ynum(0.05);
```

(27)


(28)

$$\left[x = 0.05, y_3(x) = 1.18427234235351, \frac{d}{dx} y_3(x) = -10.4293951881620 \right]$$

(28)

O gráfico da solução é mostrado a seguir:

```
> with(plots) :
> Pnum := odeplot(ynum, [x, y3(x)], 0.05 .. xf, color = black, legend = sol.numérica, linestyle = solid, numpoints = 300) :
> display(Pnum)
```



Façamos uma comparação com o resultado obtido de forma semi-automática anteriormente:

```
> Y := C[1] * (1 + (1/3) * x + (1/24) * x^2 + (1/360) * x^3 + (1/8640) * x^4) + C[2] * (-1 / (2 * x^2) + 1 / (2 * x) + (1/4) * ln(x) + 29/144) ;
```

$$Y := 0.9067070987 + 0.3023280222 x + 0.03779100277 x^2 + 0.002519400185 x^3 + 0.0001049750077 x^4 + \frac{0.0006876442120}{x^2} - \frac{0.0006876442120}{x} - 0.0003438221060 \ln(x) \quad (29)$$

Substituindo as condições iniciais temos:

```
> e1 := subs(x = 0.1, Y) = 1;
e1 := 0.9992083199 - 0.0003438221060 ln(0.1) = 1 \quad (30)
```

```
> e2 := subs(x = 0.1, diff(Y, x)) = -1;
e2 := -0.9999999989 = -1 \quad (31)
```

```
> s3 := fsolve({e1, e2})
```


s3 :=

```
> assign(s3) :
```

Façamos o gráfico da solução


```
> Pseries:= plot(Y, x = 0.5e-1 .. xf, color = blue, legend = sol.serie) :
```

```
> display(Pseries) ;
```



Façamos o gráfico simultâneo dos três resultados.

```
> display([Pmaple, Pnum, Pseries])
```



└Para a escala do problema, há uma boa concordância entre os três métodos.